

Lesson 21: Singularities and Improper Integrals

[> restart;

▼ TR_3 versus $newtoncotes_8$

```
> with(Student[Calculus1]):
ApproximateInt(f(x), x=0..1, method = newtoncotes[8],
partition=1);
TR3:= unapply((2^6*ApproximateInt(f(x),x=0..1,method=
newtoncotes[4],partition=n/4)-ApproximateInt(f(x),x=0..1,
method=newtoncotes[4],partition=n/8)))/(2^6-1),(f,n));
```

$$\frac{989}{28350} f(0) + \frac{2944}{14175} f\left(\frac{1}{8}\right) - \frac{464}{14175} f\left(\frac{1}{4}\right) + \frac{5248}{14175} f\left(\frac{3}{8}\right) - \frac{454}{2835} f\left(\frac{1}{2}\right) \\ + \frac{5248}{14175} f\left(\frac{5}{8}\right) - \frac{464}{14175} f\left(\frac{3}{4}\right) + \frac{2944}{14175} f\left(\frac{7}{8}\right) + \frac{989}{28350} f(1)$$

$$TR3 := (f, n) \rightarrow \frac{128}{2835} \frac{1}{n} \left(\sum_{i=0}^{\frac{1}{4}n-1} \left(7f\left(\frac{4i}{n}\right) + 32f\left(\frac{4\left(i+\frac{1}{4}\right)}{n}\right) \right. \right. \\ \left. \left. + 12f\left(\frac{4\left(i+\frac{1}{2}\right)}{n}\right) + 32f\left(\frac{4\left(i+\frac{3}{4}\right)}{n}\right) + 7f\left(\frac{4(i+1)}{n}\right) \right) \right) \\ - \frac{4}{2835} \frac{1}{n} \left(\sum_{i=0}^{\frac{1}{8}n-1} \left(7f\left(\frac{8i}{n}\right) + 32f\left(\frac{8\left(i+\frac{1}{4}\right)}{n}\right) + 12f\left(\frac{8\left(i+\frac{1}{2}\right)}{n}\right) \right. \right. \\ \left. \left. + 32f\left(\frac{8\left(i+\frac{3}{4}\right)}{n}\right) + 7f\left(\frac{8(i+1)}{n}\right) \right) \right) \quad (1.1)$$

One thing that's not so nice about $newtoncotes_8$ is that it has some negative coefficients. So you could have a function that's positive everywhere, but the Newton-Cotes approximation for its integral is negative. The TR rules don't have that problem: their coefficients are all positive. This implies that if $c \leq f(x) \leq d$ on the interval $a \leq x \leq b$, the TR rules will all give values between $c(b-a)$ and $d(b-a)$. Here's an example.

```
> f:= x -> exp(-100*(x-1/2)^2);
evalf(ApproximateInt(f(x),x=0..1,method=newtoncotes[8],
partition=1));
```

$$f:=x \rightarrow e^{-100\left(x-\frac{1}{2}\right)^2} \\ -0.0050586062$$

(1.2)

```
> evalf(TR3(f,8));
```

0.1528474004

(1.3)

```
> evalf(int(f(x),x=0..1));
```

0.1772453851

(1.4)

```
> ApproximateInt(f(x),x=0..1,method=newtoncotes[8],partition=1,  
output=plot);
```


An Approximation of the Integral of

$f(x) = \exp(-100*(x-1/2)^2)$

on the Interval $[0, 1]$

Using Newton Cotes' Rule of Order 8

Area: $-.5058606171e-2$



Removing Singularities

The various numerical integration methods we have considered all depend on the integrand being a "smooth" function. The error estimates for the Trapezoid and Midpoint Rules, for example, assume that f'' is continuous, while that for Simpson's Rule assumes a continuous 4th derivative. The higher-order rules require even more continuous derivatives.

Although these rules will "work" for functions that do not have so many continuous derivatives, in the sense that the limit as $n \rightarrow \infty$ is correct, they may approach that limit quite slowly. We must therefore modify our approach in dealing with such functions.

A typical function given by a closed-form expression will be nice and smooth almost everywhere in

the interval we're integrating over, but might have a finite number of "bad" points where some derivative doesn't exist. We sometimes call these **singularities**.

There are even more serious problems in dealing with **improper integrals**. Here the numerical methods we've used can't even get started (e.g. for an integral over an infinite interval, or an integrand that is undefined at some point).

Of course, some improper integrals diverge, and these are outside the reach of any numerical integration method (but we would want to detect this divergence before trying numerical methods).

Let's see what Maple does with an improper integral having a singularity.

```
> J := Int(1/sqrt(x^6+x), x=0..infinity);
```

$$J := \int_0^{\infty} \frac{1}{\sqrt{x^6+x}} dx \quad (2.1)$$

This integral is improper for two reasons:

- 1) the interval extends to ∞
- 2) the integrand goes to ∞ as $x \rightarrow 0$.

Maple can actually get an exact expression for this integral.

```
> value(J);
```

$$\frac{1}{5} \frac{\pi^{3/2} \csc\left(\frac{2}{5}\pi\right) \csc\left(\frac{1}{10}\pi\right)}{\Gamma\left(\frac{3}{5}\right) \Gamma\left(\frac{9}{10}\right)} \quad (2.2)$$

```
> evalf(%);
```

$$2.381159642 \quad (2.3)$$

But here we're interested in numerical integration. I'll use **infolevel** to take a peek at how Maple evaluates the integral numerically.

```
> infolevel[evalf]:= 1;
```

$$\text{infolevel}_{\text{evalf}} := 1 \quad (2.4)$$

```
> evalf(J);
```

```
Control: Entering NAGInt
```

```
trying d01amc (nag_ld_quad_inf)
```

```
d01amc: result=2.38115964324029328
```

$$2.381159643 \quad (2.5)$$

It used a NAG routine for doing improper integrals. That wasn't very enlightening. I'll set `Digits = 16` so it can't use NAG.

```
> Digits := 16:
```

```
evalf(J);
```

```
evalf/int/improper: integrating on interval 0 .. infinity
```

```
evalf/int/improper: applying transformation x = 1/x
```

```
evalf/int/improper: interval is 0 .. 1 for the integrand:
```

$$\frac{1}{\sqrt{x^6+x}}$$

```
and interval is 0 .. 1 for the integrand:
```

$$\frac{1}{\sqrt{\frac{1}{x^6} + \frac{1}{x} x^2}}$$

```
evalf/int/control: singularity at left end-point
evalf/int/transform: series contains {1/x^(1/2), x^(9/2), x^(19/2), x^(29/2)}
evalf/int/singleleft: applying transformation x = x^2
evalf/int/singleleft: interval is 0 .. 1. for the integrand:
```

$$\frac{2x}{\sqrt{x^2(x^{10}+1)}}$$

```
evalf/int/CreateProc: Trying makeproc
evalf/int/ccquad: n = 2 integral estimate =
1.901718361844954267
                    n = 6 integral estimate =
1.932428506967484700
evalf/int/ccquad: n = 18 integral estimate =
1.932472354636796129
                    error = .43847669311429e-4
evalf/int/ccquad: n = 54 integral estimate =
1.932472354613339933
                    error =
.1932472354613339933e-16
From ccquad, result = 1.932472354613339933 integrand evals =
55 error = .1932472354613339933e-16
evalf/int/CreateProc: Trying makeproc
evalf/int/ccquad: n = 2 integral estimate =
.4460951061426452080
                    n = 6 integral estimate =
.4486902912512757713
evalf/int/ccquad: n = 18 integral estimate =
.4486872886274276844
                    error = .30026238480869e-5
evalf/int/ccquad: n = 54 integral estimate =
.4486872886274019966
                    error =
.4486872886274019966e-17
From ccquad, result = .4486872886274019966 integrand evals =
55 error = .4486872886274019966e-17
                    2.381159643240742
```

(2.6)

```
> infolevel[evalf]:= 0;
```

```
infolevelevalf:= 0
```

(2.7)

One of the main tools Maple uses (and we will too) for dealing with singularities is transformation (change of variables).

The simplest way to transform an infinite interval to a finite one is the change of variable $t = \frac{1}{x}$, which takes $[a, \infty)$ to $(0, 1/a]$ for $a > 0$. In our example we want to use $a = 0$, but that's only a minor annoyance. One solution is to use $t = \frac{1}{x+1}$, which takes $[0, \infty)$ to $(0, 1]$. The other, which is the one Maple uses in this case, is to break up the interval into two parts, $[0, 1]$ and $[1, \infty)$, and use the transformation to take $[1, \infty)$ to $(0, 1]$, leaving the other interval alone. Since we'll also have a problem at $x = 0$ to deal with, it's a good idea to split it up, and deal with one problem at a time.

```
> J1:= Int(1/sqrt(x^6+x), x=0..1);
    J2:= Int(1/sqrt(x^6+x), x=1..infinity);
```

$$J1 := \int_0^1 \frac{1}{\sqrt{x^6+x}} dx$$

$$J2 := \int_1^{\infty} \frac{1}{\sqrt{x^6+x}} dx \quad (2.8)$$

```
> with(Student[Calculus1]):
> J2:= rhs(Rule[change,x=1/t](J2));
```

$$J2 := \int_1^0 \left(-\frac{t}{\sqrt{1+t^5}} \right) dt \quad (2.9)$$

This one should be no problem for any of our approximate integration methods. Here's our integrand (**Integrand** is a command in the **Student[Calculus1]** package):

```
> Integrand(J2);
```

$$\left[-\frac{t}{\sqrt{1+t^5}} \right] \quad (2.10)$$

It returns a list so it can work on expressions containing several integrals. We want the first element.

```
> integrand:= %[1];
```

$$integrand := -\frac{t}{\sqrt{1+t^5}} \quad (2.11)$$

```
> v50:= evalf(ApproximateInt(integrand, t=1..0, method=simpson,
    partition=50));
```

$$v50 := -0.4486872890603840 \quad (2.12)$$

ApproximateInt doesn't like integrating from 1 to 0, so we have to switch the endpoints. That introduces a minus sign.

```
> v50:= - evalf(ApproximateInt(integrand, t=0..1, method=
    simpson, partition=50));
```

$$v50 := 0.4486872890603840 \quad (2.13)$$

```
> v25:= - evalf(ApproximateInt(integrand, t=0..1, method=
```

```

simpson, partition=25));
v25 := 0.4486872955670174

```

(2.14)

Here's the error estimate for v50 according to Richardson (since Simpson's Rule is of order 4)

```

> (v50 - v25)/(2^4-1);
-4.3377556 10^-10

```

And the Richardson approximation for the integral is

```

> JR2:=(2^4*v50 - v25)/(2^4-1);
JR2 := 0.4486872886266084

```

Just to check, the actual error in v50 is (up to roundoff error)

```

> evalf(J2 - v50);
-4.329820 10^-10

```

(2.15)

and the actual error in JR2 is

```

> evalf(J2-JR2);
7.936 10^-13

```

(2.16)

Now for the other part J1. The problem here is at $x = 0$, where the denominator is 0.

```

> J1;

```

$$\int_0^1 \frac{1}{\sqrt{x^6 + x}} dx$$

(2.17)

What Maple seems to have done is a transformation $x = t^2$.

```

> J3:= rhs(Rule[change, x=t^2](J1));

```

$$J3 := \int_0^1 \frac{2}{\sqrt{t^{10} + 1}} dt$$

(2.18)

Notice that this has no problem at $t = 0$.

How did Maple know that this transformation would work? It did a series expansion. This is essentially a Taylor series, but a little more general when necessary.

```

> series(exp(x)*cos(x), x=Pi);
-e^pi - e^pi (x - pi) + 1/3 e^pi (x - pi)^3 + 1/6 e^pi (x - pi)^4 + 1/30 e^pi (x - pi)^5 + O((x - pi)^6)

```

(2.19)

`series(..., x=a)` gives us a certain number of terms in powers of $x - a$ and then $O((x - a)^n)$ to say that's what the remainder is. We can give it a third input to control to what order the series will be computed.

```

> series(exp(x)*cos(x), x=Pi, 8);
-e^pi - e^pi (x - pi) + 1/3 e^pi (x - pi)^3 + 1/6 e^pi (x - pi)^4 + 1/30 e^pi (x - pi)^5 - 1/630 e^pi (x - pi)^7 + O((x - pi)^8)

```

(2.20)

Actually the result of `series(..., x=a, n)` will not necessarily have $O((x - a)^n)$. Basically the `n` controls how the series is computed, rather than what the result will be.

```

> series((sin(x)-tan(x))/x^3, x=0, 5);

```

(2.21)

$$-\frac{1}{2} + O(x^2) \quad (2.21)$$

```
> series((sin(x)-tan(x))/x^3, x=0, 8);
```

$$-\frac{1}{2} - \frac{1}{8}x^2 - \frac{13}{240}x^4 + O(x^5) \quad (2.22)$$

The series for our integrand around $x=0$ involves non-integer powers of x (that's why I said it was a little more general than Taylor series).

```
> series(1/sqrt(x^6+x), x=0, 10);
```

$$\frac{1}{\sqrt{x}} - \frac{1}{2}x^{9/2} + \frac{3}{8}x^{19/2} + O(x^{29/2}) \quad (2.23)$$

```
> Rule[change, x = t^2](Int(g(x), x=0..1));
```

$$\int_0^1 g(x) dx = \int_0^1 2g(t^2)t dt \quad (2.24)$$

If $g(x)$ involves half-integer powers of x , $g(t^2)$ will involve integer powers of t .

And after multiplying by t , $\frac{1}{\sqrt{x}}$ becomes 1.

```
> integrand:=Integrand(J3)[1];
```

$$integrand := \frac{2}{\sqrt{t^{10} + 1}} \quad (2.25)$$

```
> v50:= evalf(ApproximateInt(integrand, t=0..1, method=simpson,
partition=50));
```

$$v50 := 1.932472361846249 \quad (2.26)$$

```
> v25:= evalf(ApproximateInt(integrand, t=0..1, method=simpson,
partition=25));
```

$$v25 := 1.932472471047547 \quad (2.27)$$

Here's Richardson's error estimate for $v50$:

```
> (v50 - v25)/(2^4-1);
```

$$-7.2800866 \cdot 10^{-9} \quad (2.28)$$

Maybe we'd like a bit more accuracy here.

```
> v100:= evalf(ApproximateInt(integrand, t=0..1, method=
simpson, partition=100));
```

$$v100 := 1.932472355064716 \quad (2.29)$$

```
> (v100 - v50)/(2^4-1);
```

$$-4.521022 \cdot 10^{-10} \quad (2.30)$$

```
> JR1:= (2^4*v100 - v50)/(2^4-1);
```

$$JR1 := 1.932472354612614 \quad (2.31)$$

So our value for the integral is

```
> JR1 + JR2;
```

$$2.381159643239222 \quad (2.32)$$

And this compares to the "exact" value:

```
> evalf(J);
```

$$2.381159643240742 \quad (2.33)$$

```
> %-%;
```

$$1.520 \cdot 10^{-12} \quad (2.34)$$

Another method for removing singularities (on a finite interval) is subtraction. Let's go back to J1.

```
> J1;
```

$$\int_0^1 \frac{1}{\sqrt{x^6+x}} dx \quad (2.35)$$

```
> integrand:= Integrand(J1)[1];
```

$$\text{integrand} := \frac{1}{\sqrt{x^6+x}} \quad (2.36)$$

```
> S:= series(integrand, x=0, 10);
```

$$S := \frac{1}{\sqrt{x}} - \frac{1}{2} x^{9/2} + \frac{3}{8} x^{19/2} + O(x^{29/2}) \quad (2.37)$$

The idea is that after subtracting off the "bad" terms (in particular the $\frac{1}{\sqrt{x}}$), the rest should be OK for something like Simpson's rule. The bad terms can be integrated exactly.

```
> badpart:= convert(S, polynom);
```

$$\text{badpart} := \frac{1}{\sqrt{x}} - \frac{1}{2} x^{9/2} + \frac{3}{8} x^{19/2} \quad (2.38)$$

```
> goodpart:= integrand - badpart;
```

$$\text{goodpart} := \frac{1}{\sqrt{x^6+x}} - \frac{1}{\sqrt{x}} + \frac{1}{2} x^{9/2} - \frac{3}{8} x^{19/2} \quad (2.39)$$

```
> badint:= int(badpart,x=0..1);
```

$$\text{badint} := \frac{599}{308} \quad (2.40)$$

```
> ApproximateInt(goodpart,x=0..1,method=simpson, partition=50);
```

$$\text{Float(undefined)} \quad (2.41)$$

Oops. The problem here is that two terms in **goodpart** are undefined at $x=0$.

```
> goodpart:= piecewise(x=0,0, goodpart);
```

$$\text{goodpart} := \begin{cases} 0 & x=0 \\ \frac{1}{\sqrt{x^6+x}} - \frac{1}{\sqrt{x}} + \frac{1}{2} x^{9/2} - \frac{3}{8} x^{19/2} & \text{otherwise} \end{cases} \quad (2.42)$$

Removing singularities by subtraction

[Another method for removing singularities (on a finite interval) is subtraction.

```
> integrand:= Integrand(J1)[1];
```

$$\text{integrand} := \frac{1}{\sqrt{x^6 + x}} \quad (3.1)$$

```
> S:= series(integrand, x=0, 10);
```

$$S := \frac{1}{\sqrt{x}} - \frac{1}{2} x^{9/2} + \frac{3}{8} x^{19/2} + O(x^{29/2}) \quad (3.2)$$

The idea is that after subtracting off the "bad" terms (in particular the $\frac{1}{\sqrt{x}}$), the rest should be OK for something like Simpson's rule. The bad terms can be integrated exactly.

```
> badpart:= convert(S, polynomial);
```

$$\text{badpart} := \frac{1}{\sqrt{x}} - \frac{1}{2} x^{9/2} + \frac{3}{8} x^{19/2} \quad (3.3)$$

```
> goodpart:= integrand - badpart;
```

$$\text{goodpart} := \frac{1}{\sqrt{x^6 + x}} - \frac{1}{\sqrt{x}} + \frac{1}{2} x^{9/2} - \frac{3}{8} x^{19/2} \quad (3.4)$$

```
> badint:= int(badpart,x=0..1);
```

$$\text{badint} := \frac{599}{308} \quad (3.5)$$

```
> ApproximateInt(goodpart,x=0..1,method=simpson, partition=50);  
Float(undefined) \quad (3.6)
```

Oops. The problem here is that two terms in **goodpart** are undefined at $x=0$.

```
> goodpart:= piecewise(x=0,0, goodpart);
```

$$\text{goodpart} := \begin{cases} 0 & x=0 \\ \frac{1}{\sqrt{x^6 + x}} - \frac{1}{\sqrt{x}} + \frac{1}{2} x^{9/2} - \frac{3}{8} x^{19/2} & \text{otherwise} \end{cases} \quad (3.7)$$

```
> goodint50:= evalf(ApproximateInt(goodpart,x=0..1,method=  
simpson, partition=50));
```

$$\text{goodint50} := -0.01233285130180981 \quad (3.8)$$

```
> v50:=evalf(badint+goodint50);
```

$$v50 := 1.932472343503385 \quad (3.9)$$

```
> evalf(J1-v50);
```

$$1.1109955 \cdot 10^{-8} \quad (3.10)$$


▼ An oscillatory integral

Here's another improper integral. In this case it's not at all obvious that it even converges. It does so because of rapid oscillation.

```
> J := Int(x*cos(x^3),x=0..infinity);
```

$$J := \int_0^{\infty} x \cos(x^3) dx$$


```
> plot(x*cos(x^3),x=0..5);
```



```
> F:=int(x*cos(x^3),x);
```

$$F := \frac{1}{6} 2^{2/3} \sqrt{\pi} \left(\frac{3}{2} \frac{2^{1/3} \sin(x^3)}{\sqrt{\pi} x} + \frac{3}{2} \frac{2^{1/3} (\cos(x^3) x^3 - \sin(x^3))}{\sqrt{\pi} x} \right. \\ \left. - \frac{1}{2} \frac{x^5 2^{1/3} \sin(x^3) \text{LommelS1}\left(\frac{1}{6}, \frac{3}{2}, x^3\right)}{\sqrt{\pi} (x^3)^{7/6}} \right. \\ \left. - \frac{3}{2} \frac{x^5 2^{1/3} (\cos(x^3) x^3 - \sin(x^3)) \text{LommelS1}\left(\frac{7}{6}, \frac{1}{2}, x^3\right)}{\sqrt{\pi} (x^3)^{13/6}} \right) \quad (4.1)$$

```
> plot(F,x=0..5);
```



In some Maple versions there's an odd "glitch" in the graph near $x = 4.3$. Sometimes plots can be improved by telling Maple to use more points, with the **numpoints** option.

```
> plot(F,x=0..5, numpoints=1000);
```


This actually is improper at both ends, even though the original J was fine at 0. To avoid the problems at $x=0$, split J into two pieces.

```
> J1 := Int(x*cos(x^3), x=0..1);
    J2 := Int(x*cos(x^3), x=1..infinity);
```

$$J1 := \int_0^1 x \cos(x^3) dx$$

$$J2 := \int_1^{\infty} x \cos(x^3) dx$$

J1 is no problem.

```
> V1 := evalf(ApproximateInt(x*cos(x^3), x=0..1, method=simpson,
    partition=50));
```

$$V1 := 0.4404076890886103$$

Now use the substitution $t = x^3$ on J2.

```
> J3 := rhs(Rule[change, t=x^3](J2));
```

$$J3 := \int_1^{\infty} \frac{1}{3} \frac{\cos(t)}{t^{1/3}} dt$$

The next trick to use is integration by parts.

```
> J4 := Rule[parts, 1/t^(1/3)/3, sin(t)](J3);
```

$$J4 := \int_1^{\infty} x \cos(x^3) dx = \lim_{t \rightarrow \infty} \frac{1}{3} \frac{\sin(t)}{t^{1/3}} - \frac{1}{3} \sin(1) - \left(\int_1^{\infty} \left(-\frac{1}{9} \frac{\sin(t)}{t^{4/3}} \right) dt \right)$$

Maple should know that limit. What's going on? When you want to see what's below the surface in a Maple expression, without the fancy typesetting, the **lprint** command is useful.

```
> lprint(%);
```

```
CALCULUS1OBJECT([6, [4], [], {x}] = Limit((1/3)*sin(t)/t^(1/3), t = infinity, left)-(1/3)*sin(1)-(Int(-(1/9)*sin(t)/t^(4/3), t = 1 .. infinity))
```

Ah! it's using an inert **Limit** rather than a **limit** that would actually find the limit.

```
> J4 := eval(rhs(J4), Limit=limit);
```

$$J4 := -\frac{1}{3} \sin(1) - \left(\int_1^{\infty} \left(-\frac{1}{9} \frac{\sin(t)}{t^{4/3}} \right) dt \right)$$

Notice that this integration by parts gave us a factor $t^{-\frac{4}{3}}$ instead of $t^{-\frac{1}{3}}$, so the integrand goes to 0 much faster as $t \rightarrow \infty$. At this point we can see that the integral converges by a comparison test:

$$\left| \frac{\sin(t)}{9 t^{\frac{4}{3}}} \right| \leq \frac{1}{9 t^{\frac{4}{3}}}, \text{ and } \int_1^{\infty} \frac{1}{9 t^{\frac{4}{3}}} dt < \infty.$$

We'll try this a few more times to make it converge faster. Actually, the **Calculus1** package's integration by parts is a bit too picky to be convenient (it wants the exact specification of the u and

v, besides having this **Limit** vs **limit** problem). There is an **IntegrationTools** package that has a **Parts** command that is easier to use, where you need only specify the expression u to be differentiated (it chooses the antiderivative of dv to use for v; that may or may not be the best choice, but you can override the choice if you want). It also has some other useful commands for manipulating integrals. We could have done the steps so far with this instead of **Student** [Calculus1].

```
> with(IntegrationTools);
```

```
[Change, CollapseNested, Combine, Expand, ExpandMultiple, Flip, GetIntegrand,
  GetOptions, GetParts, GetRange, GetVariable, Parts, Split, StripOptions]
```

```
> S:=Split(J,1);
```

$$S := \int_0^1 x \cos(x^3) dx + \int_1^\infty x \cos(x^3) dx$$

```
> J1:= op(1,S); J2:= op(2,S);
```

$$J1 := \int_0^1 x \cos(x^3) dx$$

$$J2 := \int_1^\infty x \cos(x^3) dx$$

(4.2)

```
> J3:= Change(J2,x=t^(1/3));
```

$$J3 := \int_1^\infty \frac{1}{3} \frac{\cos(t)}{t^{1/3}} dt$$

```
> J4:= Parts(J3,t^(-1/3));
```

$$J4 := -\frac{1}{3} \sin(1) - \left(\int_1^\infty \left(-\frac{1}{9} \frac{\sin(t)}{t^{4/3}} \right) dt \right)$$

(4.3)

And now for another integration by parts.

```
> J5:= Parts(J4,t^(-4/3));
```

$$J5 := -\frac{1}{3} \sin(1) + \frac{1}{9} \cos(1) + \int_1^\infty \left(-\frac{4}{27} \frac{\cos(t)}{t^{7/3}} \right) dt$$

(4.4)

Can we change variables to a proper integral now?

```
> Change(J5,t=1/u);
```

$$-\frac{1}{3} \sin(1) + \frac{1}{9} \cos(1) - \frac{4}{27} \int_0^1 \cos\left(\frac{1}{u}\right) u^{1/3} du$$

The integrand here is continuous as $u \rightarrow 0$ but its first derivative isn't:

```
> diff(cos(1/u)*u^(1/3),u);
```

$$\frac{\sin\left(\frac{1}{u}\right)}{u^{5/3}} + \frac{1}{3} \frac{\cos\left(\frac{1}{u}\right)}{u^{2/3}}$$

So we need to keep going a few more times.

```
> for k from 0 to 7 do
  J5 := Parts(J5,t^(-k-7/3))
end do;
```

$$J5 := -\frac{5}{27} \sin(1) + \frac{1}{9} \cos(1) - \left(\int_1^{\infty} \frac{28}{81} \frac{\sin(t)}{t^{10/3}} dt \right)$$

$$J5 := -\frac{5}{27} \sin(1) - \frac{19}{81} \cos(1) + \int_1^{\infty} \frac{280}{243} \frac{\cos(t)}{t^{13/3}} dt$$

$$J5 := -\frac{325}{243} \sin(1) - \frac{19}{81} \cos(1) - \left(\int_1^{\infty} \left(-\frac{3640}{729} \frac{\sin(t)}{t^{16/3}} \right) dt \right)$$

$$J5 := -\frac{325}{243} \sin(1) + \frac{3469}{729} \cos(1) + \int_1^{\infty} \left(-\frac{58240}{2187} \frac{\cos(t)}{t^{19/3}} \right) dt$$

$$J5 := \frac{55315}{2187} \sin(1) + \frac{3469}{729} \cos(1) - \left(\int_1^{\infty} \frac{1106560}{6561} \frac{\sin(t)}{t^{22/3}} dt \right)$$

$$J5 := \frac{55315}{2187} \sin(1) - \frac{1075339}{6561} \cos(1) + \int_1^{\infty} \frac{24344320}{19683} \frac{\cos(t)}{t^{25/3}} dt$$

$$J5 := -\frac{23846485}{19683} \sin(1) - \frac{1075339}{6561} \cos(1) - \left(\int_1^{\infty} \left(-\frac{608608000}{59049} \frac{\sin(t)}{t^{28/3}} \right) dt \right)$$

$$J5 := -\frac{23846485}{19683} \sin(1) + \frac{598929949}{59049} \cos(1) + \int_1^{\infty} \left(-\frac{17041024000}{177147} \frac{\cos(t)}{t^{31/3}} \right) dt$$

```
> J6 := Change(J5,t=1/u);
```

$$J6 := -\frac{23846485}{19683} \sin(1) + \frac{598929949}{59049} \cos(1) - \frac{17041024000}{177147} \int_0^1 \cos\left(\frac{1}{u}\right) u^{25/3} du$$

Maple objects introduced in this lesson

Limit

numpoints option for **plot**

IntegrationTools package

Split (in **IntegrationTools**)

Change (in **IntegrationTools**)

Parts (in **IntegrationTools**)